

# Lesson 17: Integration of Rational Functions

```
[> restart;
```

## Partial fractions: the "Math 101" method

We were looking at how to integrate a rational function. After an initial division step, which produces a polynomial part of the integral, we were left with a rational function  $\frac{p}{q}$  where the degree of  $p$  is less than the degree of  $q$  (this is not the original  $p$ , but I'll call it  $p$  now). In our example:

```
> p := 13*x^3 + 16*x^2 + 5*x + 1;
q := x^4 - 6*x^2 - 8*x - 3;
f := p/q;
```

$$f := \frac{13x^3 + 16x^2 + 5x + 1}{x^4 - 6x^2 - 8x - 3}$$

We factored the denominator  $q$ .

```
> factor(q);
```

$$(x - 3)(x + 1)^3 \quad (1.1)$$

Now  $\frac{p}{q}$  is supposed to be decomposed into partial fractions: a sum of the following form:

```
> parfrac := p/q = a/(x-3) + b/(x+1) + c/(x+1)^2 + d/(x+1)^3;
```

$$\text{parfrac} := \frac{13x^3 + 16x^2 + 5x + 1}{x^4 - 6x^2 - 8x - 3} = \frac{a}{x-3} + \frac{b}{x+1} + \frac{c}{(x+1)^2} + \frac{d}{(x+1)^3}$$

We need to solve for the constants  $a, b, c, d$  that make this equation true. If we clear away the denominators, we get an equation for polynomials:

```
> normal(parfrac*q);
```

$$13x^3 + 16x^2 + 5x + 1 = ax^3 + 3ax^2 + 3ax + a + bx^3 - bx^2 - 5bx - 3b + cx^2 - 2cx - 3c + dx - 3d$$

The coefficient of each power of  $x$  on the left must match the coefficient of each power of  $x$  on the right.

```
> eqns := {seq(coeff(rhs(%), x, n) = coeff(lhs(%), x, n), n=0..3)};
eqns := {a + b = 13, c + 3a - b = 16, -2c + 3a - 5b + d = 5, -3d - 3c + a - 3b = 1}
```

These are four linear equations in the four unknown coefficients  $a, b, c, d$ , which Maple can solve:

```
> solve(eqns);
```

$$\left\{ a = \frac{511}{64}, b = \frac{321}{64}, c = -\frac{47}{16}, d = \frac{1}{4} \right\}$$

Plug these back in to the partial fraction form.

```
> eval(rhs(parfrac), %);
```

$$\frac{511}{64(x-3)} + \frac{321}{64(x+1)} - \frac{47}{16(x+1)^2} + \frac{1}{4(x+1)^3}$$

Of course, we know how to integrate each of those terms. The result (added to the integral of the polynomial part) is our antiderivative.

> **F = int(%, x);**

$$F = \frac{511}{64} \ln(x-3) + \frac{321}{64} \ln(x+1) + \frac{47}{16(x+1)} - \frac{1}{8(x+1)^2}$$

The main problem in general with this approach is that we might not be able to factor the denominator. Let's see some examples where the denominator doesn't factor quite so nicely. Sometimes there's no way around it: you just have to write an answer involving RootOfs.

> **int(p/((x^3 + x + 1)\*(x^2+1)^2), x);**

$$\sum_{R=\text{RootOf}(\_Z^3 + \_Z + 1)} \frac{(8 - 8\_R - 28\_R^2) \ln(x - \_R)}{3\_R^2 + 1} + \frac{1}{4} \frac{-30x + 16}{x^2 + 1} + \frac{1}{2} \arctan(x) + 14 \ln(x^2 + 1)$$

> **int(p/(x^6 - 4\*x^4 + 3\*x^3 + 22\*x^2 - 28), x);**

$$\frac{1}{1187} \sum_{R=\text{RootOf}(\_Z^3 - 2\_Z^2 + 7)} \frac{(9694 + 2011\_R - 1002\_R^2) \ln(x - \_R)}{3\_R^2 - 4\_R} + \frac{1}{1187} \sum_{R=\text{RootOf}(\_Z^3 + 2\_Z^2 - 4)} \frac{(5709 + 1997\_R + 1002\_R^2) \ln(x - \_R)}{3\_R^2 + 4\_R}$$

> **int(p/(x^3 + x + 1)^2, x);**

$$13 \left( \sum_{R=\text{RootOf}(\_Z^3 + \_Z + 1)} \frac{\ln(x - \_R)}{3\_R^2 + 1} \right) + \frac{-\frac{144}{31} - \frac{172}{31}x + \frac{32}{31}x^2}{x^3 + x + 1} + \frac{8}{31} \sum_{R=\text{RootOf}(31\_Z^3 - 7079\_Z + 80259)} \_R \ln \left( x - \frac{9114}{1993129} \_R^2 - \frac{172453}{1993129} \_R + \frac{1387484}{1993129} \right)$$

But sometimes it turns out you can avoid solving nasty polynomials.

## Partial Fractions in Maple: the Hurwitz-Ostrogradsky method

What Maple does next after the division step is called the Hurwitz-Ostrogradsky method, which produces any rational part of the integral. I'll use the same numerator  $p$  in our example.

> **q:=expand((x^2+1)\*(x^2+x+1)^2); J:= Int(p/q, x); value(J);**

$$q := x^6 + 2x^5 + 4x^4 + 4x^3 + 4x^2 + 2x + 1$$

$$J := \int \frac{13x^3 + 16x^2 + 5x + 1}{x^6 + 2x^5 + 4x^4 + 4x^3 + 4x^2 + 2x + 1} dx$$

$$-4 \ln(x^2 + x + 1) - \frac{74}{9} \sqrt{3} \arctan\left(\frac{1}{3} (2x + 1) \sqrt{3}\right) + 4 \ln(x^2 + 1) + 15 \arctan(x)$$

$$+ \frac{1}{3} \frac{20x + 13}{x^2 + x + 1}$$

We need the following useful fact:

**Any two polynomials  $a(x)$  and  $b(x)$  have a greatest common divisor  $c(x)$ : this divides both  $a(x)$  and  $b(x)$ , and every polynomial that does that also divides  $c(x)$ .**

"Greatest" here is in the sense of "highest degree".

You could get the greatest common divisor by looking at the common factors of  $a(x)$  and  $b(x)$ , but there's an efficient way of doing it without factoring. The command is called **gcd**.

```
> gcd(q, (3*x+2)*(x^2+1));
```

$$x^2 + 1 \tag{2.1}$$

Notice that any factor that occurs in  $q$  to a power  $> 1$  also divides  $q'$ , e.g.

$D(a^2 b) = 2 a D(a) b + a^2 D(b)$ . So taking the gcd of  $q$  and  $q'$  will capture those factors, but nothing that occurs only to the power 1.

```
> r:= gcd(q, diff(q,x)); s:= normal(q/r^2);
```

$$r := x^2 + x + 1$$
$$s := x^2 + 1$$

So now (with our assumption that there is no higher exponent than a square) we can say  $q = r^2 s$  where  $s$  has no squared (or higher power) factors. We say  $s$  is **squarefree**.

Now I'd like to write  $\frac{p}{r^2 s} = (a/r)' + \frac{b}{r s}$ , where  $a$  and  $b$  are polynomials:  $a$  should have lower

degree than  $r$  and  $b$  should have lower degree than  $r s$ .

Expand this out with the Quotient Rule:

$$\frac{p}{r^2 s} = \frac{r a' - a r'}{r^2} + \frac{b}{r s}$$

Multiply by  $r^2 s$ :

$$p = r a' s - a r' s + b r$$

In our example,  $a$  should have degree at most 1 and  $b$  should have degree at most 3, so take  $a = a_0 + a_1 x$  and  $b = b_0 + b_1 x + b_2 x^2 + b_3 x^3$ .

```
> A:= a[0] + a[1]*x;
```

```
 B:= b[0] + b[1]*x + b[2]*x^2 + b[3]*x^3;
```

```
 p = r * diff(A,x) * s - A*diff(r,x)*s + B*r;
```

$$A := a_0 + a_1 x$$
$$B := b_0 + b_1 x + b_2 x^2 + b_3 x^3$$

$$13 x^3 + 16 x^2 + 5 x + 1 = (x^2 + x + 1) a_1 (x^2 + 1) - (a_0 + a_1 x) (2 x + 1) (x^2 + 1) + (b_0 + b_1 x + b_2 x^2 + b_3 x^3) (x^2 + x + 1)$$

Expand this out, look at coefficients of equal powers of  $x$ , and you have six equations in six unknowns.

```
> expand(%);
```

$$13 x^3 + 16 x^2 + 5 x + 1 = b_1 x + b_3 x^3 + b_2 x^2 - a_1 x^4 - 2 a_0 x - 2 a_0 x^3 - a_0 x^2 + b_0 x^2 + b_0 x + b_1 x^3 + b_1 x^2 + b_2 x^4 + b_2 x^3 + b_3 x^5 + b_3 x^4 - a_0 + a_1 + b_0$$

```
> eqns := {seq(coeff(lhs(%), x, n) = coeff(rhs(%), x, n), n = 0..5)};
eqns := {0 = b3, 0 = -a1 + b2 + b3, 1 = -a0 + a1 + b0, 5 = b1 - 2 a0 + b0, 13 = b3 - 2 a0 + b1
        + b2, 16 = b2 - a0 + b0 + b1}
```

```
> s := solve(eqns);
```

$$S := \left\{ a_0 = \frac{13}{3}, a_1 = \frac{20}{3}, b_0 = -\frac{4}{3}, b_1 = 15, b_2 = \frac{20}{3}, b_3 = 0 \right\}$$

So the rational part of the antiderivative is

```
> rationalpart := eval(A/r, S);
```

$$\text{rationalpart} := \frac{\frac{13}{3} + \frac{20}{3}x}{x^2 + x + 1}$$

and the part we still have to integrate is

```
> todo := eval(B/(r*s), S);
```

$$\text{todo} := \frac{-\frac{4}{3} + 15x + \frac{20}{3}x^2}{(x^2 + x + 1)(x^2 + 1)}$$

I'll call the numerator and denominator here  $p$  and  $q$  again.

```
> p := numer(todo); q := denom(todo);
```

$$p := -4 + 45x + 20x^2$$

$$q := 3(x^2 + x + 1)(x^2 + 1)$$

where there are no more repeated roots.

## Partial fractions in Maple: Finding the logarithmic part

The next point: suppose  $\frac{p}{q} = \sum_j \frac{a_j}{x - r_j}$

where  $r_j$  are the roots of  $q$  (and there are no repeated roots). Then  $a_j = \frac{p(r_j)}{q'(r_j)}$ .

How does this work?

$$q = c \prod_i (x - r_i)$$

$$q' = \sum_j c \prod_{i \neq j} (x - r_i)$$

$$q'(r_j) = c \prod_{i \neq j} (r_j - r_i)$$

$$p = \sum_j \frac{a_j q}{x - r_j} = \sum_j a_j c \prod_{i \neq j} (x - r_i)$$

$$p(r_j) = a_j c \prod_{i \neq j} (r_j - r_i) = a_j q'(r_j)$$

The trouble is, I don't want to actually calculate the roots  $r_j$  if I can avoid it. Fortunately, I can avoid it.

Well,  $q$  and  $p - a_j q'$  have a common root  $r_j$ . Now we have a way of finding when polynomials have a common root, namely the resultant.

```
> R:= resultant(q, p - z*diff(q,x), x);
      R := 27 (1801 + 216 z + 27 z^2) (2601 - 288 z + 36 z^2) (3.1)
```

This should be 0 when  $z = \frac{p(r_j)}{q'(r_j)}$  for at least one of the roots  $r_j$  of  $q$ .

```
> A := {solve(R)};
      A := {4 - 15/2 I, 4 + 15/2 I, -4 - 37/9 I sqrt(3), -4 + 37/9 I sqrt(3)} (3.2)
```

That is the list of all the different coefficients  $a_j$  that we could have. Each might belong to more than one root  $r_j$ .

Now  $r_j$  is a root of  $q$  and of  $p - a_j q'$ , so it's a root of  $G_j = \gcd(q, p - a_j q')$ . Nothing can be a root of  $G_j$  that is not one of the  $r_k$  with the same coefficient. So (up to a constant factor)  $G_j$  is the product of  $x - r_k$  for all those roots  $r_k$  that have coefficient  $a_j$ . The point here is that I can calculate  $G_j$  without calculating the individual roots.

```
> for ii from 1 to 4 do
      G[ii]:= gcd(q, p - A[ii]*diff(q,x))
end do;

      G1 := x - I
      G2 := x + I
      G3 := 1/2 + x + 1/2 I sqrt(3)
      G4 := x + 1/2 - 1/2 I sqrt(3) (3.3)
```

Now  $\ln(G_j)$  is a constant plus the sum of  $\ln(x - r_k)$  for those roots  $r_k$  where the coefficient is  $a_j$ .

The derivative of  $\ln(G_j)$  is the sum of  $\frac{1}{x - r_k}$  for those roots. So the term for those roots in the antiderivative is  $a_j \ln(G_j)$ .

```
> logterms := add(A[j]*ln(G[j]), j=1..4);
logterms := (4 - 15/2 I) ln(x - I) + (4 + 15/2 I) ln(x + I) + (-4 - 37/9 I sqrt(3)) ln(1/2
      + x + 1/2 I sqrt(3)) + (-4 + 37/9 I sqrt(3)) ln(x + 1/2 - 1/2 I sqrt(3)) (3.4)
```

To check that this works:

```
> diff(logterms,x) = todo;
```

$$\frac{4 - \frac{15}{2} I}{x - I} + \frac{4 + \frac{15}{2} I}{x + I} + \frac{-4 - \frac{37}{9} I\sqrt{3}}{\frac{1}{2} + x + \frac{1}{2} I\sqrt{3}} + \frac{-4 + \frac{37}{9} I\sqrt{3}}{x + \frac{1}{2} - \frac{1}{2} I\sqrt{3}} \quad (3.5)$$

$$= \frac{-\frac{4}{3} + 15x + \frac{20}{3} x^2}{(x^2 + x + 1)(x^2 + 1)}$$

> normal(lhs(%)-rhs(%));

$$0 \quad (3.6)$$

The only other thing to mention is that these expressions in complex logarithms can be expressed in terms of real quantities. The `evalc` function tries to express a complex expression using real and imaginary parts.

> evalc(logterms) assuming x::real;

$$4 \ln(x^2 + 1) + \frac{15}{2} \arctan(-1, x) - \frac{15}{2} \arctan(1, x) - 4 \ln\left(\left(\frac{1}{2} + x\right)^2 + \frac{3}{4}\right) \quad (3.7)$$

$$+ \frac{37}{9} \sqrt{3} \arctan\left(\frac{1}{2} \sqrt{3}, \frac{1}{2} + x\right) - \frac{37}{9} \sqrt{3} \arctan\left(-\frac{1}{2} \sqrt{3}, \frac{1}{2} + x\right)$$

$$+ I\left(4 \arctan(-1, x) + 4 \arctan(1, x) - 4 \arctan\left(\frac{1}{2} \sqrt{3}, \frac{1}{2} + x\right) - 4 \arctan\left(-\frac{1}{2} \sqrt{3}, \frac{1}{2} + x\right)\right)$$

> int(todo, x);

$$-4 \ln(x^2 + x + 1) - \frac{74}{9} \sqrt{3} \arctan\left(\frac{1}{3} (2x + 1) \sqrt{3}\right) + 4 \ln(x^2 + 1) + 15 \arctan(x) \quad (3.8)$$


> simplify(%-%) assuming x::real;

$$-\frac{74}{9} \sqrt{3} \arctan\left(\frac{1}{3} (2x + 1) \sqrt{3}\right) + 15 \arctan(x) - \frac{15}{2} \arctan(-1, x) + \frac{15}{2} \arctan(1, x) \quad (3.9)$$

$$- \frac{37}{9} \sqrt{3} \arctan\left(\frac{1}{2} \sqrt{3}, \frac{1}{2} + x\right) + \frac{37}{9} \sqrt{3} \arctan\left(-\frac{1}{2} \sqrt{3}, \frac{1}{2} + x\right)$$

$$- 4 I \arctan(-1, x) - 4 I \arctan(1, x) + 4 I \arctan\left(\frac{1}{2} \sqrt{3}, \frac{1}{2} + x\right) + 4 I \arctan\left(-\frac{1}{2} \sqrt{3}, \frac{1}{2} + x\right)$$

> plot([Re(%), Im(%)], x=-10..10);



To sum up, here's how Maple integrates a rational function  $\frac{p}{q}$  :

Polynomial part:

The polynomial part is the integral of the polynomial  $quo(p, q, x)$ .

Replace  $p$  by  $rem(p, q, x)$  and proceed to the next part.

Rational part (in the case where the highest power of any factor is 2):

$$\text{Let } r = gcd(q, q'), s = \frac{q}{r^2}$$

Find polynomials  $a$  with lower degree than  $r$  and  $b$  with lower degree than  $r s$  so that  $p = r a' s - a r' s + b r$ .

The rational part is  $\frac{a}{r}$ .

Replace  $q$  by  $r s$  and  $p$  by  $b$  and proceed to the next part.

Logarithmic part:

Let  $R = resultant(q, p - z q', x)$  (a polynomial in  $z$ ).

Let  $a_1, \dots, a_k$  be the roots of  $R$ .

For each  $j$  from 1 to  $k$  let  $G_j = gcd(q, p - a_j q')$ .

The logarithmic part is

$$\sum_{j=1}^k a_j \ln(G_j).$$

## Maple objects introduced in this lesson

gcd